

Modelo del orden de lectura de periódicos

Equipo 4: Hamilton Gómez, Verónica Mendoza, Mariana Uribe

1. Presentación del problema

A continuación, se plantea un modelo matemático para resolver el siguiente problema:

Cuatro colegas de la carrera Ingeniería Matemática comparten un apartamento y se levantan temprano para asistir a las clases matinales, pero les gusta antes leer los periódicos que reciben. A las 6.00 am todas las mañanas tienen en la puerta del apartamento cuatro periódicos representativos de diferentes regiones del país (El Tiempo, El Colombiano, El Espectador y El País). Desayunan a las 7.00 am (a más tardar), disponiendo pues de una hora para ojear los distintos diarios. Los tiempos (en minutos) que cada uno de los estudiantes dedica a cada uno de los periódicos que leen aparecen en la siguiente tabla:

Estudiante/P eriodico	El Tiempo	El Colombiano	El Espectador	El País
Estudiante 1	20	5	-	15
Estudiante 2	10	10	10	10
Estudiante 3	-	15	15	10
Estudiante 4	5	10	20	5

Tabla 1: tiempo que cada estudiante emplea leyendo cada periódico

Se pretende dar un orden en el que los estudiantes deben leer los periódicos para que queden libres lo antes posible.

El problema se resolvió usando el software Gurobi con AMPL.

2. Modelación del problema

Índices:

i : Número de periódicos: }
 k : Número de periódicos: } $i, k = 1, 2, 3, 4.$

j : Número de estudiantes: }
 l : Número de estudiantes: } $j, l = 1, 2, 3, 4.$

Datos del problema:

t_{ij} : tiempo que tarda el estudiante j en leer el periódico i .

Variables de decisión:

X_{ij} : tiempo en el que el estudiante j inicia la lectura del periódico i .

$W_{ikj} = \begin{cases} 1 & \text{Si el periódico } i \text{ es leído antes que el periódico } k \text{ por el estudiante } j; \\ 0 & \text{en otro caso.} \end{cases}$

$Y_{jli} = \begin{cases} 1 & \text{Si el estudiante } j \text{ lee antes que el estudiante } l \text{ el periódico } i; \\ 0 & \text{en otro caso.} \end{cases}$

T : tiempo máximo en el que alguien termina de leer todos los periódicos correspondientes.

Función Objetivo:

Minimizar $Z = T$

Restricciones:

$$(1) \quad X_{ij} + t_{ij} \leq X_{kj} + M(1 - W_{ikj}) \quad \forall j, \forall i \neq k$$

$$(2) \quad X_{kj} + t_{kj} \leq X_{ij} + MW_{ikj} \quad \forall j, \forall i \neq k$$

$$(3) \quad T \geq X_{ij} + t_{ij} \quad \forall i, \forall j$$

$$(4) \quad X_{ij} + t_{ij} \leq X_{il} + M(1 - Y_{jli}) \quad \forall i, \forall j \neq l$$

$$(5) \quad X_{il} + t_{il} \leq X_{ij} + MY_{jli} \quad \forall i, \forall j \neq l$$

$$(6) \quad X_{ij} \geq 0$$

$$(7) \quad T \geq 0$$

Las restricciones (1) y (2) establecen que el tiempo en que un estudiante inicia la lectura de un primer periódico más el tiempo que tarda, debe ser menor a el tiempo en que inició la lectura de otro periódico garantizando así que ningún estudiante lea dos periódicos a la vez. para estas restricciones también se establece un valor M grande que condiciona el cumplimiento de una de las dos.

la restricción (3) se establece que el tiempo total de lectura es el máximo tiempo de terminación entre los estudiantes, es decir, es el máximo tiempo desde que inicia la lectura más el tiempo que le toma leer.

En la restricción (4) se plantea que, si un estudiante lee un periódico antes que otro estudiante, el tiempo en que el primer estudiante inicia la lectura más el tiempo que tarda leyendo, debe ser menor al tiempo en el que el siguiente estudiante inicia a leerlo; esto verifica que un mismo periódico nunca será leído por dos estudiantes en ningún instante de tiempo, pues es necesario que el primer estudiante termine. La restricción (5) plantea la misma situación para el caso contrario, esto es, que el segundo estudiante lee antes que el primer estudiante. Además, para ambas restricciones se establece un valor M

suficientemente grande que admita la ocurrencia de una sola restricción pues solo se puede dar un caso.

Las restricciones (6) y (7) garantizan que no haya variables negativas para cumplir con el modelo real.

3. Representación del problema en AMPL

a. Fichero de datos

```
#conjunto de datos para los estudiantes y periódicos a leer

set PERIODICOS:= 1 2 3 4; #Donde 1 es El Tiempo, 2 es El Colombiano
                        #      3 es El Espectador, 4 es El País

set ESTUDIANTES:= 1 2 3 4;

#parámetro que expresa el tiempo que un estudiante demora leyendo un periódico
param time:

      1      2      3      4:=
1    20      5      0      15
2    10      10     10     10
3     0      15     15     10
4     5      10     20     5;
```

b. Fichero del modelo

```
'
set PERIODICOS;
set ESTUDIANTES;

#parametros

#time es l parámetro tij del modelo
param time{PERIODICOS,ESTUDIANTES};

#variables

#time_start es la variabe Xij del modelo
var time_start{PERIODICOS,ESTUDIANTES}>=0;

#prec_estudiantes es la variable Yjli del modelo
var prec_estudiantes{ESTUDIANTES,ESTUDIANTES,PERIODICOS}binary;

# prec_periodicos es la variable Wikj del modelo
var prec_periodicos{PERIODICOS,PERIODICOS,ESTUDIANTES}binary;

#total_time es la variable T del modelo
var total_time>=0;

#función objetivo

minimize t_time: total_time;
```

```

#restricciones

#Las restricciones [1] y [2] verifican que un estudiante j no lea dos periódicos al
#mismo tiempo, estableciendo un orden de lectura para los periódicos por estudiante.

#[1]
subject to prec_est1{i in PERIODICOS,k in PERIODICOS,j in ESTUDIANTES: i<>k}:
    (time_start[i,j] + time[i,j]) <= time_start[k,j]+ 60*(1-prec_periodicos[i,k,j]);

#[2]
subject to prec_est2{i in PERIODICOS,k in PERIODICOS,j in ESTUDIANTES: i<>k}:
    (time_start[k,j] + time[k,j]) <= time_start[i,j]+ 60*prec_periodicos[i,k,j];

#[3]
#Establece el mayor valor que puede demorar un estudiante leyendo todos los periódicos
subject to T_mayor{i in PERIODICOS,j in ESTUDIANTES}:
    total_time>= time_start[i,j] + time[i,j];

#Las restricciones [4] y [5] verifican que dos estudiantes no lean un periódico j al
#mismo tiempo, estableciendo un orden de lectura para los estudiantes por periódico.

#[4]
subject to prec_per1{j in ESTUDIANTES,l in ESTUDIANTES,i in PERIODICOS: j<>l}:
    time_start[i,j]+time[i,j]<= time_start[i,l]+ 60*(1-prec_estudiantes[j,l,i]);

#[5]
subject to prec_per2{j in ESTUDIANTES,l in ESTUDIANTES,i in PERIODICOS: j<>l}:
    time_start[i,l]+time[i,l]<= time_start[i,j]+ 60*prec_estudiantes[j,l,i];

```

4. Referencias

- [1] Carrasco,F.(2016). Modelado y resolución de problemas de secuenciación de tareas. Aplicación a una empresa de fabricación de electrodos. Escuela Técnica Superior de Ingeniería. Sevilla, España. Recuperado el 23 de Octubre de 2019 de <https://bit.ly/32nwlzm>
- [2] Gómez, C.[Cristhian Gomez].(2014, Mayo 20). AMPL Programación Lineal - Modelo Algebraico[Archivo de video]. Recuperado de <https://bit.ly/2pB7MHD>